Некоторые точные решения

К лекции 4 (2024)

1 мКдФ

1.1 мКдФ⁺

In[1]:=
```
eq[v_] := -D[v, t] + D[v, x, x, x] + 6 v^2 D[v, x]
Simplify[eq[k/Cosh[k x + k^3 t + d]]]
```

Out[2]= 0

1.2 мКдФ⁻

In[3]:=
```
u = a Tanh[a x - 2 a^3 t + d];
Simplify[D[u, t] - D[D[u, x, x] - 2 u^3, x]]

Plot[u /. {a → 1, d → 0, t → 1}, {x, -7, 7},
 PlotStyle → {Blue},
 PlotRange → {-1.15, 1.15},
 BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
 ImageSize → 400,
 AxesLabel → {"x", "u"}]
```

Out[4]= 0

Out[5]=



Проиллюстрируем один способ рекуррентного построения решений (при помощи преобразований Бэклнда; будет объяснено позже на лекциях).

In[6]:=
```
F[a_, d_] := -a Tanh[a x + 4 a^3 t + d]
u = F[a, d]
Simplify[-D[u, t] + D[u, x, x, x] - 6 (u^2 - a^2) D[u, x]]
```

Out[7]= $-a\,\mathrm{Tanh}[d + 4 a^3 t + a x]$

Out[8]= 0

In[9]:=
```
F[a1_, d1_, a2_, d2_] = -F[a1, d1] - (a2^2 - a1^2)/(F[a1, d1] - F[a2, d2])
u = F[a1, d1, a2, d2]
Together[TrigToExp[-D[u, t] + D[u, x, x, x] - 6 (u^2 - a2^2) D[u, x]]]
```

Out[9]= $a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] - \frac{-a1^2 + a2^2}{-a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] + a2\,\mathrm{Tanh}[d2 + 4 a2^3 t + a2 x]}$

Out[10]=

$a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] - \frac{-a1^2 + a2^2}{-a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] + a2\,\mathrm{Tanh}[d2 + 4 a2^3 t + a2 x]}$

Out[11]=

0

In[12]:=
```
F[a1_, d1_, a2_, d2_, a3_, d3_] := -F[a1, d1, a2, d2] - (a3^2 - a2^2)/(F[a1, d1, a2, d2] - F[a1, d1, a3, d3])

u = F[a1, d1, a2, d2, a3, d3]
Together[TrigToExp[-D[u, t] + D[u, x, x, x] - 6 (u^2 - a3^2) D[u, x]]]
```

Out[13]=

$$-a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] + \frac{-a1^2 + a2^2}{-a1\,\mathrm{Tanh}[d1 + 4 a1^3 t + a1 x] + a2\,\mathrm{Tanh}[d2 + 4 a2^3 t + a2 x]} - \frac{-a2^2 + a3^2}{-\frac{-a1^2+a2^2}{-a1\,\mathrm{Tanh}[d1+4 a1^3 t+a1 x]+a2\,\mathrm{Tanh}[d2+4 a2^3 t+a2 x]} + \frac{-a1^2+a3^2}{-a1\,\mathrm{Tanh}[d1+4 a1^3 t+a1 x]+a3\,\mathrm{Tanh}[d3+4 a3^3 t+a3 x]}}$$

Out[14]=

0

In[15]:=
```
Tanh[x + ⅈ π / 2]
```

Out[15]=

Coth[x]

In[16]:=
```
Manipulate[
 Plot[F[a1, d1, a2, d2 + ⅈ π / 2, a3, d3] /. t → tt, {x, -40, 40},
  PlotStyle → {Blue},
  PlotRange → {-2.15, 2.15},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 500,
  AxesLabel → {"x", "u"}],
 {{a1, 0.6}, 0, 2, Appearance → "Labeled"},
 {{a2, 0.8}, 0, 2, Appearance → "Labeled"},
 {{a3, 1}, 0, 2, Appearance → "Labeled"},
 {{d1, 0}, -5, 5, Appearance → "Labeled"},
 {{d2, 0}, -5, 5, Appearance → "Labeled"},
 {{d3, 3}, -5, 5, Appearance → "Labeled"},
 {{tt, 0}, -12, 12, Appearance → "Labeled"}]
```

Out[16]=



2 Bsq $u_{tt} = \pm(u_{xx} + 3u^2)_{xx}$

2.1 $u_{tt} = (u_{xx} + 3u^2)_{xx}$

В отличие от КдФ, солитоны могут двигаться в любую сторону

In[17]:=
```
eq[u_] := -D[u, t, t] + D[D[u, x, x] + 3 u^2, x, x]

uBq1[x_, t_, k_, d_] = 2 k^2 / Cosh[k x + 2 k^2 t + d]^2

uBq2[x_, t_, k_, d_] = 2 k^2 / Cosh[k x - 2 k^2 t + d]^2

Simplify[{eq[uBq1[x, t, k, d]], eq[uBq2[x, t, k, d]]}]
```

Out[18]=

$2 k^2 \text{Sech}[d + 2 k^2 t + k x]^2$

Out[19]=

$2 k^2 \text{Sech}[d - 2 k^2 t + k x]^2$

Out[20]=

{0, 0}

In[21]:=
```
Manipulate[
 Plot[uBq1[x, t, k, 0], {x, -20, 20},
  PlotStyle → {Blue},
  PlotRange → {-2.15, 2.15},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{k, 1}, -2, 2, Appearance → "Labeled"},
 {{t, 0}, -2, 2, Appearance → "Labeled"}]
```

Out[21]=



2.2 $u_{tt} = -\left(u_{xx} + 3\,u^2\right)_{xx}$

Тригонометрическое решение

In[22]:=
```
Clear[u1, u2]
eq[u_] := D[u, t, t] + D[D[u, x, x] + 3 u^2, x, x]

uBqm1[x_, t_, k_, d_] = -(2 k^2)/(Cos[k x + 2 k^2 t + d]^2)

uBqm2[x_, t_, k_, d_] = -(2 k^2)/(Cos[k x - 2 k^2 t + d]^2)

Simplify[{eq[uBqm1[x, t, k, d]], eq[uBqm2[x, t, k, d]]}]
```

Out[24]=

$-2\,k^2\,\mathrm{Sec}\left[d + 2\,k^2\,t + k\,x\right]^2$

Out[25]=

$-2\,k^2\,\mathrm{Sec}\left[d - 2\,k^2\,t + k\,x\right]^2$

Out[26]=

{0, 0}

In[27]:=
```
Manipulate[
 Plot[uBqm1[x, t, 1, 0], {x, -20, 20},
  PlotStyle → {Blue},
  PlotRange → {-5.15, 0.15},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{t, 0}, -2, 2, Appearance → "Labeled"}]
```

Out[27]=



3 NLS$^{\pm}$ $\quad i\,q_t = q_{xx} \pm 2\,|q|^2 q, \quad q = u + i\,v$

Общие формулы для быстроубывающего n-солитонного решения и некоторые специальные случаи (раздел 3.4):

• В.Е. Захаров, А.Б. Шабат.Точная теория двумерной самофокусировки и одномерной автомодуляции волн в нелинейных средах. ЖЭТФ 61:1 (1971) 118-134.

См. также
• M.J. Ablowitz, D.J. Kaup, A.C. Newell, H. Segur. Nonlinear evolution equations of physical significance. Phys. Rev. Let. 31:2 (1973) 125-127.
• В.Е. Захаров, С.В. Манаков, С.П. Новиков, Л.П. Питаевский. Теория солитонов. Метод обратной задачи, М.: Наука, 1980.
• М. Абловиц, Х. Сигур. Солитоны и метод обратной задачи. М.: Мир, 1987.
• Н.Н. Ахмедиев, А. Анкевич. Солитоны. М.: Физматлит, 2003.

3.1 Классические симметрии

Однопараметрические группы преобразований, не меняющие систему

$$i\,q_t = q_{xx} + 2\,q^2\,p, \quad -i\,p_t = p_{xx} + 2\,q\,p^2$$

In[28]:=
```
Clear[q, p, a, σ]
NLSru = {
   Derivative[0, 1][q][x_, t_] :→ -ⅈ (Derivative[2, 0][q][x, t] + 2 q[x, t]^2 p[x, t]),
   Derivative[0, 1][p][x_, t_] :→ ⅈ (Derivative[2, 0][p][x, t] + 2 q[x, t] p[x, t]^2)};

NLS[Q_, P_] := Expand[{-ⅈ D[Q, t] + D[Q, x, x] + 2 Q^2 P, ⅈ D[P, t] + D[P, x, x] + 2 Q P^2} /. NLSru]
```

Сдвиги по x и по t

In[31]:=
```
NLS[q[x + a, t], p[x + a, t]]
NLS[q[x, t + a], p[x, t + a]]
```

Out[31]=
```
{0, 0}
```

Out[32]=
```
{0, 0}
```

Сдвиг аргумента

In[33]:=
```
NLS[Exp[ⅈ a] q[x, t], Exp[-ⅈ a] p[x, t]]
```

Out[33]=
```
{0, 0}
```

Скейлинг

In[34]:=
```
NLS[Exp[a] q[Exp[a] x, Exp[2 a] t], Exp[a] p[Exp[a] x, Exp[2 a] t]]
```

Out[34]=
```
{0, 0}
```

Преобразование Галилея

In[35]:=
```
NLS[Exp[ⅈ (a x + a^2 t)] q[x + 2 a t, t], Exp[-ⅈ (a x + a^2 t)] p[x + 2 a t, t]]
```

Out[35]=
```
{0, 0}
```

Аналогичные замены для системы

$$q_t = q_{xx} + 2\,q^2\,p,\quad -p_t = p_{xx} + 2\,q\,p^2$$

In[36]:=
```
NLSru = {
   Derivative[0, 1][q][x_, t_] :> Derivative[2, 0][q][x, t] + 2 q[x, t]^2 p[x, t],
   Derivative[0, 1][p][x_, t_] :> -(Derivative[2, 0][p][x, t] + 2 q[x, t] p[x, t]^2)};

NLS[Q_, P_] := Expand[{-D[Q, t] + D[Q, x, x] + 2 Q^2 P, D[P, t] + D[P, x, x] + 2 Q P^2} /. NLSru]

NLS[q[x + a, t], p[x + a, t]]

NLS[q[x, t + a], p[x, t + a]]

NLS[Exp[a] q[x, t], Exp[-a] p[x, t]]

NLS[Exp[a] q[Exp[a] x, Exp[2 a] t], Exp[a] p[Exp[a] x, Exp[2 a] t]]

NLS[Exp[a x + a^2 t] q[x + 2 a t, t], Exp[-a x - a^2 t] p[x + 2 a t, t]]
```

Out[38]=
{0, 0}

Out[39]=
{0, 0}

Out[40]=
{0, 0}

Out[41]=
{0, 0}

Out[42]=
{0, 0}

3.2 Бозе-конденсат

In[43]:=
```
cc[f_] := f /. {Complex[a_, b_] :> Complex[a, -b]}
cri[f_] := ComplexExpand[ReIm[f]]

qNLS[x_, t_, a_, b_, c_, s_] = b Exp[ⅈ (a x + (a^2 - 2 s b^2) t + c)];
q = %
cc[q]
cri[q]

Together[-ⅈ D[q, t] + D[q, x, x] + 2 s cc[q] q^2]
```

Out[46]=
$b\, e^{i\,(c+(a^2-2 b^2 s)\, t+a x)}$

Out[47]=
$b\, e^{-i\,(c+(a^2-2 b^2 s)\, t+a x)}$

Out[48]=
{b Cos[c + (a^2 - 2 b^2 s) t + a x], b Sin[c + (a^2 - 2 b^2 s) t + a x]}

Out[49]=
0

In[50]:=
```
qq = qNLS[x, t, 0.3, 1, 0, 1]

Plot3D[Evaluate[cri[qq][[1]]], {x, -10, 10}, {t, -8, 8},
 PlotRange → {-2, 2},
 PlotPoints → 80, Mesh → False,
 ImageSize → 500,
 BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
 AxesLabel → {"x", "t", "Re u"}
]
```

Out[50]=

$e^{i(-1.91t+0.3x)}$

Out[51]=



3.3 NLS⁺: солитон огибающей

Envelope soliton

Частный случай при $b = 0$: бризер (breather)

Солитон КдФ содержит 2 параметра – амплитуду и фазовый сдвиг. Амплитуду можно сделать равной 1 растяжением, сдвиг убирается сдвигом x. То есть, фактически, параметров нет (если речь именно об одном солитоне).

В солитоне НУШ есть 4 параметра, из них c, d можно сдвинуть в 0, один из параметров a или b можно растянуть в 1, но второй остается свободным.

In[52]:=

```
qNLS[x_, t_, a_, b_, c_, d_] = ComplexExpand[TrigToExp[a Exp[ⅈ (b x + (b^2 - a^2) t + c)] / Cosh[a (x + 2 b t) + d]]];

q = %

Simplify[-ⅈ D[q, t] + D[q, x, x] + 2 cc[q] q^2]

{u, v} = cri[q]
```

Out[53]=

$$\frac{2\,a\,\mathrm{Cos}\left[c+\left(-a^{2}+b^{2}\right)t+b\,x\right]}{e^{-d-a\,(2\,b\,t+x)}+e^{d+a\,(2\,b\,t+x)}}+\frac{2\,i\,a\,\mathrm{Sin}\left[c+\left(-a^{2}+b^{2}\right)t+b\,x\right]}{e^{-d-a\,(2\,b\,t+x)}+e^{d+a\,(2\,b\,t+x)}}$$

Out[54]=

0

Out[55]=

$$\left\{\frac{2\,a\,\mathrm{Cos}\left[c+\left(-a^{2}+b^{2}\right)t+b\,x\right]}{e^{-d-a\,(2\,b\,t+x)}+e^{d+a\,(2\,b\,t+x)}},\ \frac{2\,a\,\mathrm{Sin}\left[c+\left(-a^{2}+b^{2}\right)t+b\,x\right]}{e^{-d-a\,(2\,b\,t+x)}+e^{d+a\,(2\,b\,t+x)}}\right\}$$

In[56]:=

```
plotNLS1[a_, b_, c_, d_, t_] := Module[{u, v},
  {u, v} = cri[qNLS[x, t, a, b, c, d]];
  Plot[{u, v, Sqrt[u^2 + v^2]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-1.15, 1.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

esPr1 = plotNLS1[1, 12, 0, 0, 0]
```

Out[57]=



In[58]:=
```
Manipulate[
 plotNLS1[a, b, c, d, t],
 {{a, 1}, 0, 2, Appearance → "Labeled"},
 {{b, 0}, -30, 30, Appearance → "Labeled"},
 {{c, 0}, -3, 3, Appearance → "Labeled"},
 {{d, 0}, -3, 3, Appearance → "Labeled"},
 {{t, 0}, -10, 10, Appearance → "Labeled"}]
```

Out[58]=



In[59]:=
```
Plot3D[Re[qNLS[x, t, 1, 3, 0, 0]], {x, -10, 10}, {t, -3, 3},
 PlotRange → {-1, 1},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Re u"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[59]=



3.4 NLS⁺: двухсолитонное решение

Детерминантная формула для n-солитонного решения

$$A = (A_{jk})_1^n, \quad A_{jk} = \frac{c_k}{\lambda_k - \bar{\lambda}_j} \exp\left(-2\, i\, (\bar{\lambda}_j\, x - 2\, \lambda_k^2\, t)\right)$$

$$M = I + \bar{A}\, A$$

$$M_1 = \begin{pmatrix} M & E_1 \\ E_2 & 0 \end{pmatrix}, \quad E_1 = (\exp(2\, i\, \lambda_1\, x),\ \ldots,\ \exp(2\, i\, \lambda_n\, x))^t, \quad E_2 = \left(c_1 \exp\left(4\, i\, \lambda_1^2\, t\right),\ \ldots,\ c_n \exp\left(4\, i\, \lambda_n^2\, t\right)\right)$$

$$q = -2 \det M_1 / \det M$$

3.4.1 Решение общего вида

Вронскианные формулы могут быть выведены примерно так же, как для КдФ. Однако, они более сложные – 2-солитонное решение содержит 8 вещественных параметров против 4 у КдФ. Возможны нетривиальные вырождения и предельные случаи.

1-солитонное решение

In[60]:=

```
Clear[a, b, c, d, λ, μ]

n = 1;
A = Table[c[k] Exp[-2 I (μ[j] x - 2 λ[k]^2 t)] / (λ[k] - μ[j]), {j, 1, n}, {k, 1, n}];
M = IdentityMatrix[n] + (cc[A] /. {λ → μ, μ → λ, c → d, d → c}).A;

MM = Table[
   If[j < n + 1,
    If[k < n + 1, M[[j, k]], Exp[2 I λ[j] x]],
    If[k < n + 1, c[k] Exp[4 I λ[k]^2 t], 0]],
   {j, 1, n + 1}, {k, 1, n + 1}];

q = -2 Det[MM] / Det[M]
cq = cc[q] /. {λ → μ, μ → λ, c → d, d → c};

Together[-I D[q, t] + D[q, x, x] + 2 cq q^2]
```

Out[65]=

$$\frac{2\, e^{2 i x \lambda[1]+4 i t \lambda[1]^2}\, c[1]}{1+\frac{e^{-2 i \left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i \left(x \lambda[1]-2 t \mu[1]^2\right)}\, c[1]\, d[1]}{(\lambda[1]-\mu[1])\,(-\lambda[1]+\mu[1])}}$$

Out[67]=

0

2-солитонное решение

In[68]:=

```
n = 2;
A = Table[c[k] Exp[-2 I (μ[j] x - 2 λ[k]^2 t)] / (λ[k] - μ[j]), {j, 1, n}, {k, 1, n}];
M = IdentityMatrix[n] + (cc[A] /. {λ → μ, μ → λ, c → d, d → c}).A;

MM = Table[
   If[j < n + 1,
    If[k < n + 1, M[[j, k]], Exp[2 I λ[j] x]],
    If[k < n + 1, c[k] Exp[4 I λ[k]^2 t], 0]],
   {j, 1, n + 1}, {k, 1, n + 1}];

q = -2 Det[MM] / Det[M]
cq = cc[q] /. {λ → μ, μ → λ, c → d, d → c};

Together[-I D[q, t] + D[q, x, x] + 2 cq q^2]
```

Out[72]=

$$-\Bigg(\Bigg(2\,\Bigg(-e^{2 i x \lambda[1]+4 i t \lambda[1]^2}\,c[1]-e^{2 i x \lambda[2]+4 i t \lambda[2]^2}\,c[2]-$$
$$\frac{e^{2 i x \lambda[2]+4 i t \lambda[2]^2-2 i\left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i\left(x \lambda[1]-2 t \mu[1]^2\right)}\,c[1]\times c[2]\times d[1]}{(\lambda[1]-\mu[1])\,(-\lambda[1]+\mu[1])}+$$
$$\frac{e^{4 i t \lambda[1]^2+2 i x \lambda[2]-2 i\left(-2 t \lambda[2]^2+x \mu[1]\right)+2 i\left(x \lambda[1]-2 t \mu[1]^2\right)}\,c[1]\times c[2]\times d[1]}{(\lambda[2]-\mu[1])\,(-\lambda[1]+\mu[1])}+$$
$$\frac{e^{2 i x \lambda[1]+4 i t \lambda[2]^2-2 i\left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i\left(x \lambda[2]-2 t \mu[1]^2\right)}\,c[1]\times c[2]\times d[1]}{(\lambda[1]-\mu[1])\,(-\lambda[2]+\mu[1])}-$$
$$\frac{e^{2 i x \lambda[1]+4 i t \lambda[1]^2-2 i\left(-2 t \lambda[2]^2+x \mu[1]\right)+2 i\left(x \lambda[2]-2 t \mu[1]^2\right)}\,c[1]\times c[2]\times d[1]}{(\lambda[2]-\mu[1])\,(-\lambda[2]+\mu[1])}-$$
$$\frac{e^{2 i x \lambda[2]+4 i t \lambda[2]^2-2 i\left(-2 t \lambda[1]^2+x \mu[2]\right)+2 i\left(x \lambda[1]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[2]}{(\lambda[1]-\mu[2])\,(-\lambda[1]+\mu[2])}+$$
$$\frac{e^{4 i t \lambda[1]^2+2 i x \lambda[2]-2 i\left(-2 t \lambda[2]^2+x \mu[2]\right)+2 i\left(x \lambda[1]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[2]}{(\lambda[2]-\mu[2])\,(-\lambda[1]+\mu[2])}+$$
$$\frac{e^{2 i x \lambda[1]+4 i t \lambda[2]^2-2 i\left(-2 t \lambda[1]^2+x \mu[2]\right)+2 i\left(x \lambda[2]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[2]}{(\lambda[1]-\mu[2])\,(-\lambda[2]+\mu[2])}-$$
$$\frac{e^{2 i x \lambda[1]+4 i t \lambda[1]^2-2 i\left(-2 t \lambda[2]^2+x \mu[2]\right)+2 i\left(x \lambda[2]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[2]}{(\lambda[2]-\mu[2])\,(-\lambda[2]+\mu[2])}\Bigg)\Bigg)\Big/$$
$$\Bigg(1+\frac{e^{-2 i\left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i\left(x \lambda[1]-2 t \mu[1]^2\right)}\,c[1]\times d[1]}{(\lambda[1]-\mu[1])\,(-\lambda[1]+\mu[1])}+$$
$$\frac{e^{-2 i\left(-2 t \lambda[2]^2+x \mu[1]\right)+2 i\left(x \lambda[2]-2 t \mu[1]^2\right)}\,c[2]\times d[1]}{(\lambda[2]-\mu[1])\,(-\lambda[2]+\mu[1])}+$$
$$\frac{e^{-2 i\left(-2 t \lambda[1]^2+x \mu[2]\right)+2 i\left(x \lambda[1]-2 t \mu[2]^2\right)}\,c[1]\times d[2]}{(\lambda[1]-\mu[2])\,(-\lambda[1]+\mu[2])}+$$
$$\frac{e^{-2 i\left(-2 t \lambda[2]^2+x \mu[1]\right)+2 i\left(x \lambda[2]-2 t \mu[1]^2\right)-2 i\left(-2 t \lambda[1]^2+x \mu[2]\right)+2 i\left(x \lambda[1]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[1]\times d[2]}{(\lambda[2]-\mu[1])\,(-\lambda[2]+\mu[1])\,(\lambda[1]-\mu[2])\,(-\lambda[1]+\mu[2])}-$$
$$\frac{e^{-2 i\left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i\left(x \lambda[2]-2 t \mu[1]^2\right)-2 i\left(-2 t \lambda[2]^2+x \mu[2]\right)+2 i\left(x \lambda[1]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[1]\times d[2]}{(\lambda[1]-\mu[1])\,(-\lambda[2]+\mu[1])\,(\lambda[2]-\mu[2])\,(-\lambda[1]+\mu[2])}-$$
$$\frac{e^{-2 i\left(-2 t \lambda[2]^2+x \mu[1]\right)+2 i\left(x \lambda[1]-2 t \mu[1]^2\right)-2 i\left(-2 t \lambda[1]^2+x \mu[2]\right)+2 i\left(x \lambda[2]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[1]\times d[2]}{(\lambda[2]-\mu[1])\,(-\lambda[1]+\mu[1])\,(\lambda[1]-\mu[2])\,(-\lambda[2]+\mu[2])}+$$
$$\frac{e^{-2 i\left(-2 t \lambda[2]^2+x \mu[2]\right)+2 i\left(x \lambda[2]-2 t \mu[2]^2\right)}\,c[2]\times d[2]}{(\lambda[2]-\mu[2])\,(-\lambda[2]+\mu[2])}+$$
$$\frac{e^{-2 i\left(-2 t \lambda[1]^2+x \mu[1]\right)+2 i\left(x \lambda[1]-2 t \mu[1]^2\right)-2 i\left(-2 t \lambda[2]^2+x \mu[2]\right)+2 i\left(x \lambda[2]-2 t \mu[2]^2\right)}\,c[1]\times c[2]\times d[1]\times d[2]}{(\lambda[1]-\mu[1])\,(-\lambda[1]+\mu[1])\,(\lambda[2]-\mu[2])\,(-\lambda[2]+\mu[2])}\Bigg)\Bigg)$$

Out[74]=

0

Отметим, что команда Simplify приводит формулу для q к более компактному виду, но в результате время проверки очень сильно увеличивается

In[75]:=
```
q = Simplify[-2 Det[MM] / Det[M]]
cq = cc[q] /. {λ → μ, μ → λ, c → d, d → c};

(* Simplify[-I D[q,t]+D[q,x,x]+2cq q^2] *)
```

Out[75]=

$$\left(2\left(-e^{2 i \lambda[1] (x+2 t \lambda[1])} c[1]-e^{2 i \lambda[2] (x+2 t \lambda[2])} c[2]+\frac{e^{2 i x (\lambda[1]+\lambda[2]-\mu[1])+4 i t (\lambda[1]^2+\lambda[2]^2-\mu[1]^2)} c[1] \times c[2] \times d[1] (\lambda[1]-\lambda[2])^2}{(\lambda[1]-\mu[1])^2 (\lambda[2]-\mu[1])^2}+\frac{e^{2 i x (\lambda[1]+\lambda[2]-\mu[2])+4 i t (\lambda[1]^2+\lambda[2]^2-\mu[2]^2)} c[1] \times c[2] \times d[2] (\lambda[1]-\lambda[2])^2}{(\lambda[1]-\mu[2])^2 (\lambda[2]-\mu[2])^2}\right)\right) \Big/ \left(-1+\frac{e^{2 i (\lambda[1]-\mu[1]) (x+2 t (\lambda[1]+\mu[1]))} c[1] \times d[1]}{(\lambda[1]-\mu[1])^2}+\frac{e^{2 i (\lambda[2]-\mu[1]) (x+2 t (\lambda[2]+\mu[1]))} c[2] \times d[1]}{(\lambda[2]-\mu[1])^2}+\frac{e^{2 i (\lambda[1]-\mu[2]) (x+2 t (\lambda[1]+\mu[2]))} c[1] \times d[2]}{(\lambda[1]-\mu[2])^2}+\frac{e^{2 i (\lambda[2]-\mu[2]) (x+2 t (\lambda[2]+\mu[2]))} c[2] \times d[2]}{(\lambda[2]-\mu[2])^2}-\left(e^{2 i x (\lambda[1]+\lambda[2]-\mu[1]-\mu[2])+4 i t (\lambda[1]^2+\lambda[2]^2-\mu[1]^2-\mu[2]^2)} c[1] \times c[2] \times d[1] \times d[2] (\lambda[1]-\lambda[2])^2 (\mu[1]-\mu[2])^2\right) \Big/ \left((\lambda[1]-\mu[1])^2 (\lambda[2]-\mu[1])^2 (\lambda[1]-\mu[2])^2 (\lambda[2]-\mu[2])^2\right)\right)$$

3.4.2 NLS$^+$: спаренный солитон (2 солитона с одинаковой скоростью и разной амплитудой)

Для уравнения КдФ, скорости солитонов обязаны быть разными. В НУШ они разные в случае общего положения, но при этом можно выбрать параметры и таким образом, чтобы скорости совпали. Для такого 2-солитонного решения перейдем в подвижную с.к., применяя преобразование Галилея, и сделаем скорость равной 0.

In[77]:=
```
qNLSd[x_, t_, a_, b_, c_] =
   2 (a^2 - b^2) (a Exp[-ⅈ a^2 t] Cosh[b (x + c)] - b Exp[-ⅈ b^2 t] Cosh[a x]) /
   ((a - b)^2 Cosh[a x + b (x + c)] + (a + b)^2 Cosh[a x - b (x + c)] - 4 a b Cos[(a^2 - b^2) t]);
q = %

Simplify[Together[-ⅈ D[q, t] + D[q, x, x] + 2 cc[q] q^2]]
```

Out[78]=

$$\frac{2 (a^2-b^2) \left(-b\, e^{-i b^2 t} \mathrm{Cosh}[a x]+a\, e^{-i a^2 t} \mathrm{Cosh}[b (c+x)]\right)}{-4 a b \mathrm{Cos}[(a^2-b^2) t]+(a+b)^2 \mathrm{Cosh}[a x-b (c+x)]+(a-b)^2 \mathrm{Cosh}[a x+b (c+x)]}$$

Out[79]=

0

In[80]:=

```
plotNLSd[a_, b_, c_, t_] := Module[{u, v},
  {u, v} = cri[qNLSd[x, t, a, b, c]];
  Plot[{u, v, Sqrt[u^2 + v^2]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-5.15, 5.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

Manipulate[
 plotNLSd[a, b, c, t],
 {{a, 1}, -5, 5, Appearance → "Labeled"},
 {{b, 2}, -5, 5, Appearance → "Labeled"},
 {{c, -3}, -5, 5, Appearance → "Labeled"},
 {{t, 0}, -10, 10, Appearance → "Labeled"}]
```

Out[81]=



In[82]:=
```
Plot3D[Re[qNLSd[x, t, 1.2, 2, 3]], {x, -10, 10}, {t, -3, 3},
 PlotRange → {-2, 2},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Re u"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[82]=



In[83]:=
```
Plot3D[Re[qNLSd[x, t, 1, 1.2, -0.4]], {x, -20, 20}, {t, -6, 6},
 PlotRange → {-2, 2},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Re u"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[83]=



3.4.3 NLS$^+$: полурациональный солитон (предельный случай предыдущего решения)

Берем предыдущее решение

In[84]:=
```
qNLSd[x_, t_, a_, b_, c_] =
  2 (a^2 - b^2) (a Exp[-i a^2 t] Cosh[b (x + c)] - b Exp[-i b^2 t] Cosh[a x]) / 
  ((a - b)^2 Cosh[a x + b (x + c)] + (a + b)^2 Cosh[a x - b (x + c)] - 4 a b Cos[(a^2 - b^2) t]);
q = %
```

Out[85]=

$$\frac{2\left(a^2-b^2\right)\left(-b\, e^{-i b^2 t}\,\mathrm{Cosh}[a x]+a\, e^{-i a^2 t}\,\mathrm{Cosh}[b (c+x)]\right)}{-4 a b\,\mathrm{Cos}\left[\left(a^2-b^2\right) t\right]+(a+b)^2\,\mathrm{Cosh}[a x-b (c+x)]+(a-b)^2\,\mathrm{Cosh}[a x+b (c+x)]}$$

Заменяем $c = \delta(a - b)$, где $\delta = \text{const}$, и переходим к пределу $b \to a$.
При этом числитель и знаменатель равны 0, применяем правило Лопиталя (два раза).

In[86]:=
```
numden = {Numerator[q], Denominator[q]} /. c → δ (a - b);
% /. b → a
```

Out[87]=
```
{0, 0}
```

In[88]:=
```
dnumden = D[numden, b]
% /. b → a
```

Out[88]=
```
{-4 b (-b ⅇ^(-ⅈ b^2 t) Cosh[a x] + a ⅇ^(-ⅈ a^2 t) Cosh[b (x + (a - b) δ)]) + 2 (a^2 - b^2)
    (-ⅇ^(-ⅈ b^2 t) Cosh[a x] + 2 ⅈ b^2 ⅇ^(-ⅈ b^2 t) t Cosh[a x] + a ⅇ^(-ⅈ a^2 t) (x + (a - b) δ - b δ) Sinh[b (x + (a - b) δ)]),
 -4 a Cos[(a^2 - b^2) t] + 2 (a + b) Cosh[a x - b (x + (a - b) δ)] - 2 (a - b) Cosh[a x + b (x + (a - b) δ)] -
  8 a b^2 t Sin[(a^2 - b^2) t] + (a + b)^2 (-x - (a - b) δ + b δ) Sinh[a x - b (x + (a - b) δ)] +
  (a - b)^2 (x + (a - b) δ - b δ) Sinh[a x + b (x + (a - b) δ)]}
```

Out[89]=
```
{0, 0}
```

In[90]:=
```
ddnumden = D[dnumden, b];
% /. b → a
qq = %[[1]] / %[[2]]
```

Out[91]=
```
{-8 a (-ⅇ^(-ⅈ a^2 t) Cosh[a x] + 2 ⅈ a^2 ⅇ^(-ⅈ a^2 t) t Cosh[a x] + a ⅇ^(-ⅈ a^2 t) (x - a δ) Sinh[a x]),
 2 + 16 a^4 t^2 + 4 a^2 (-x + a δ)^2 + 2 Cosh[2 a x]}
```

Out[92]=

$$-\frac{8 a \left(-e^{-i a^2 t} \operatorname{Cosh}[a x]+2 i a^2 e^{-i a^2 t} t \operatorname{Cosh}[a x]+a e^{-i a^2 t} (x-a \delta) \operatorname{Sinh}[a x]\right)}{2+16 a^4 t^2+4 a^2 (-x+a \delta)^2+2 \operatorname{Cosh}[2 a x]}$$

In[93]:=
```
qNLSr[x_, t_, a_, δ_] =
  -(8 a (-ⅇ^(-ⅈ a^2 t) Cosh[a x] + 2 ⅈ a^2 ⅇ^(-ⅈ a^2 t) t Cosh[a x] + a ⅇ^(-ⅈ a^2 t) (x - a δ) Sinh[a x])) /
    (2 + 16 a^4 t^2 + 4 a^2 (-x + a δ)^2 + 2 Cosh[2 a x]);

q = %

Simplify[Together[-ⅈ D[q, t] + D[q, x, x] + 2 cc[q] q^2]]
```

Out[94]=

$$-\frac{8 a \left(-e^{-i a^2 t} \operatorname{Cosh}[a x]+2 i a^2 e^{-i a^2 t} t \operatorname{Cosh}[a x]+a e^{-i a^2 t} (x-a \delta) \operatorname{Sinh}[a x]\right)}{2+16 a^4 t^2+4 a^2 (-x+a \delta)^2+2 \operatorname{Cosh}[2 a x]}$$

Out[95]=
```
0
```

Получилось 2 солитона одинаковой амплитуды и с логарифмической разностью скорости

In[96]:=

```
plotNLSr[a_, d_, t_] := Module[{u, v},
  {u, v} = cri[qNLSr[x, t, a, d]];
  Plot[{u, v, Sqrt[u² + v²]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-5.15, 5.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

Manipulate[
 plotNLSr[a, d, t],
 {{a, 3}, -5, 5, Appearance → "Labeled"},
 {{d, 0}, -30, 30, Appearance → "Labeled"},
 {{t, 0}, -100, 100, Appearance → "Labeled"}]
```

Out[97]=



In[98]:=
```
Plot3D[Re[qNLSr[x, t, 1, 3]], {x, -10, 10}, {t, -40, 40},
 PlotRange → {-2, 2},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Re u"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[98]=



3.5 NLS$^+$: солитон Перегрина (рациональный солитон)

• D.H. Peregrine. Water waves, nonlinear Schrödinger equations and their solutions. J. Austral. Math. Soc. Ser. B 25 (1983) 16-43.

In[99]:=

```
qNLSP[x_, t_] = Exp[-2 ⅈ t] (1 - (4 (1 - 4 ⅈ t))/(1 + 4 x^2 + 16 t^2));
q = %
cri[q]

Simplify[Together[-ⅈ D[q, t] + D[q, x, x] + 2 cc[q] q^2]]
```

Out[100]=

$$e^{-2 i t} \left(1 - \frac{4 (1 - 4 i t)}{1 + 16 t^2 + 4 x^2}\right)$$

Out[101]=

$$\left\{\text{Cos}[2 t] - \frac{4 \text{Cos}[2 t]}{1 + 16 t^2 + 4 x^2} + \frac{16 t \text{Sin}[2 t]}{1 + 16 t^2 + 4 x^2}, \frac{16 t \text{Cos}[2 t]}{1 + 16 t^2 + 4 x^2} - \text{Sin}[2 t] + \frac{4 \text{Sin}[2 t]}{1 + 16 t^2 + 4 x^2}\right\}$$

Out[102]=

0

In[103]:=

```
plotNLSP[t_] := Module[{u, v},
  {u, v} = cri[qNLSP[x, t]];
  Plot[{u, v, Sqrt[u^2 + v^2]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-3.15, 3.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

Manipulate[
 plotNLSP[t],
 {{t, 0}, -10, 10, Appearance → "Labeled"}]
```

Out[104]=



In[105]:=

```
Plot3D[Abs[qNLSP[x, t]], {x, -6, 6}, {t, -2, 2},
 PlotRange → {-1, 4},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "|q|"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[105]=



In[106]:=

```
Plot3D[Re[qNLSP[x, t]], {x, -6, 6}, {t, -2, 2},
 PlotRange → {-3, 4},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Re q"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}]

Plot3D[Im[qNLSP[x, t]], {x, -6, 6}, {t, -2, 2},
 PlotRange → {-3, 4},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "Im q"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}]
```

Out[106]=



Out[107]=



3.6 NLS$^+$: солитон Ма

(или солитон на фоне конденсата)

• E.A. Kuznetsov. Solitons in a parametrically unstable plasma. Sov. Phys. Dokl. 236 (1977) 575-577.
• T. Kawata, H. Inoue. Inverse scattering method for the nonlinear evolution equations under nonvanishing conditions. J. Phys. Soc. Japan 44:5 (1978) 1722-1729.
• Y.-C. Ma. The perturbed plane-wave solution of the cubic Schrödinger equation. Stud. Appl. Math. 60

(1979) 43-58.

In[108]:=

```
qNLSMa[x_, t_, a_, b_] =
  a Exp[-2 ⅈ a^2 t] (1 + (2 b (b Cos[4 a^2 b c t] - ⅈ Sqrt[1 + b^2] Sin[4 a^2 b c t]))/(c Cosh[2 a b x] + Cos[4 a^2 b c t])) /. c → Sqrt[1 + b^2];
q = %
Simplify[Together[-ⅈ D[q, t] + D[q, x, x] + 2 cc[q] q^2]]
```

Out[109]=

$$a\, e^{-2 i a^2 t} \left(1 + \frac{2 b \left(b \cos\left[4 a^2 b \sqrt{1+b^2}\, t\right] - i \sqrt{1+b^2} \sin\left[4 a^2 b \sqrt{1+b^2}\, t\right]\right)}{\cos\left[4 a^2 b \sqrt{1+b^2}\, t\right] + \sqrt{1+b^2} \cosh[2 a b x]}\right)$$

Out[110]=

0

In[111]:=

```
plotNLSMa[t_, a_, b_] := Module[{u, v},
  {u, v} = cri[qNLSMa[x, t, a, b]];
  Plot[{u, v, Sqrt[u^2 + v^2]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-3.15, 3.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

Manipulate[
 plotNLSMa[t, a, b],
 {{a, 1}, -5, 5, Appearance → "Labeled"},
 {{b, 1}, -5, 5, Appearance → "Labeled"},
 {{t, 0}, -10, 10, Appearance → "Labeled"}]
```

Out[112]=



In[113]:=

```
Plot3D[Re[qNLSMa[x, t, 1, 1]], {x, -6, 6}, {t, -8, 8},
 PlotRange → {-4, 4},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "|q|"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[113]=



3.7 NLS⁻: темный солитон

In[114]:=

```
qNLSdark[x_, t_, a_, b_] = -b Tanh[b (x + 2 a t)] Exp[ⅈ (2 b^2 - a^2) t + ⅈ a (x + 2 a t)];

q = %
cri[q]

Simplify[Together[-ⅈ D[q, t] + D[q, x, x] - 2 cc[q] q^2]]
```

Out[115]=

$$-b\, e^{i\left(-a^2+2 b^2\right) t+i a (2 a t+x)} \operatorname{Tanh}[b (2 a t+x)]$$

Out[116]=

$$\left\{-\frac{b \operatorname{Cos}\left[\left(-a^2+2 b^2\right) t+a (2 a t+x)\right] \operatorname{Sinh}[2 b (2 a t+x)]}{1+\operatorname{Cosh}[2 b (2 a t+x)]},\right.$$
$$\left.-\frac{b \operatorname{Sin}\left[\left(-a^2+2 b^2\right) t+a (2 a t+x)\right] \operatorname{Sinh}[2 b (2 a t+x)]}{1+\operatorname{Cosh}[2 b (2 a t+x)]}\right\}$$

Out[117]=

0

In[118]:=

```
plotNLSdark[t_, a_, b_] := Module[{u, v},
  {u, v} = cri[qNLSdark[x, t, a, b]];
  Plot[{u, v, Sqrt[u^2 + v^2]}, {x, -7, 7},
   PlotStyle → {Blue, Green, Red},
   PlotRange → {-3.15, 3.15},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 400,
   AxesLabel → {"x", None},
   PlotLegends → LineLegend[{Blue, Green, Red}, {"Re q", "Im q", "|q|"}]
  ]]

Manipulate[
 plotNLSdark[t, a, b],
 {{a, 1}, -5, 5, Appearance → "Labeled"},
 {{b, 1}, -5, 5, Appearance → "Labeled"},
 {{t, 0}, -10, 10, Appearance → "Labeled"}]
```

Out[119]=



In[120]:=

```
Plot3D[Re[qNLSdark[x, t, 1, 2]], {x, -6, 6}, {t, -3, 3},
 PlotRange → {-4, 4},
 PlotPoints → 200,
 Mesh → False,
 ImageSize → 500,
 AxesLabel → {"x", "t", "|q|"},
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}
]
```

Out[120]=



4 Sin-Гордон $u_{xy} = \sin u$

4.1 Приведение к рациональному виду

Вычисления удобнее делать для уравнения в рациональной форме

$$v_{xy} = \frac{v}{1+v^2}\left(2\, v_x\, v_y + 1 - v^2\right), \quad v = \tan\frac{u}{4}.$$

In[121]:=

```
Clear[u, v]
u = 4 ArcTan[v[x, y]]
FullSimplify[D[u, x, y] - Sin[u] /.
  {D[v[x, y], x, y] → v[x, y]/(1 + v[x, y]^2) (2 D[v[x, y], x] × D[v[x, y], y] + 1 - v[x, y]^2)}]
```

Out[122]=

4 ArcTan[v[x, y]]

Out[123]=

0

1-солитонное решение

In[124]:=

```
v = v1 = c Exp[a x + y / a]
Factor[D[v, x, y] - v/(1 + v^2) (2 D[v, x] × D[v, y] + 1 - v^2)]
u = 4 ArcTan[v]
FullSimplify[D[u, x, y] - Sin[u]]
```

Out[124]=

$c\, e^{a x+\frac{y}{a}}$

Out[125]=

0

Out[126]=

$4\, \mathrm{ArcTan}\left[c\, e^{a x+\frac{y}{a}}\right]$

Out[127]=

0

2-солитонное решение

In[128]:=

```
v = v2 = ((a1 + a2) (c2 Exp[a2 x + y / a2] - c1 Exp[a1 x + y / a1]))/((a1 - a2) (1 + c1 c2 Exp[(a1 + a2) x + (1 / a1 + 1 / a2) y]))
Simplify[D[v, x, y] - v/(1 + v^2) (2 D[v, x] × D[v, y] + 1 - v^2)]
```

Out[128]=

$$\frac{(a1+a2)\left(-c1\, e^{a1\, x+\frac{y}{a1}}+c2\, e^{a2\, x+\frac{y}{a2}}\right)}{(a1-a2)\left(1+c1\, c2\, e^{(a1+a2)\, x+\left(\frac{1}{a1}+\frac{1}{a2}\right) y}\right)}$$

Out[129]=

0

Здесь параметры не обязательно вещественны. Выбирая их по разному, получаем разные типы решений.

4.2 Кинк

In[130]:=

```
Plot3D[4 ArcTan[v1] /. {a → 0.5, c → 1}, {x, -12, 12}, {y, -8, 8},
 PlotRange → {0, 2 π},
 PlotPoints → 50,
 Mesh → False,
 ColorFunction → "Aquamarine",
 AxesLabel → {"x", "y", "u"},
 ImageSize → 500,
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}]
```

Out[130]=



4.3 Кинк-антикинк

In[131]:=

```
Plot3D[4 ArcTan[v2] /. {a1 → 0.5, a2 → 1, c1 → 1, c2 → 1}, {x, -12, 12}, {y, -8, 8},
 PlotRange → {-2.1 π, 2.1 π},
 PlotPoints → 50,
 Mesh → False,
 ColorFunction → "Aquamarine",
 AxesLabel → {"x", "y", "u"},
 ImageSize → 500,
 BaseStyle → {FontSize → 12, FontFamily → "Times New Roman"}]
```

Out[131]=



In[132]:=

```
Manipulate[
 Plot[4 ArcTan[v2] /. {a1→0.5, a2→1, c1→1, c2→1, y→yy}, {x, -50, 50},
  PlotRange→{-2.1 π, 2.1 π},
  PlotPoints→50,
  PlotStyle→Blue,
  AxesLabel→{"x", "u"},
  ImageSize→500,
  BaseStyle→{FontSize→14, FontFamily→"Times New Roman"}],
 {{yy, 0}, -10, 10}
]
```

Out[132]=



4.4 Кинк-кинк

In[133]:=

```
Plot3D[4 ArcTan[v2] /. {a1 → 0.5, a2 → -1, c1 → 1, c2 → 1}, {x, -12, 12}, {y, -8, 8},
 PlotRange → {-2.1 π, 2.1 π},
 PlotPoints → 50,
 Mesh → False,
 ColorFunction → "Aquamarine",
 AxesLabel → {"x", "y", "u"},
 ImageSize → 500,
 BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"}]
```

Out[133]=



In[134]:=

```
Manipulate[
 Plot[4 ArcTan[v2] /. {a1 → 0.5, a2 → -1, c1 → 1, c2 → 1, y → yy}, {x, -50, 50},
  PlotRange → {-2.1 π, 2.1 π},
  PlotPoints → 50,
  PlotStyle → Blue,
  AxesLabel → {"x", "u"},
  ImageSize → 500,
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"}],
 {{yy, -15}, -20, 20}
]
```

Out[134]=



4.5 Бризер

Пусть параметры комплексно сопряжены:

In[135]:=

```
v2
v2 /. {a1 → a + I b, a2 → a - I b, c1 → c + I d, c2 → c - I d}
v = Factor[ComplexExpand[%]]
Simplify[D[v, x, y] - v/(1 + v^2) (2 D[v, x] × D[v, y] + 1 - v^2)]
```

Out[135]=

$$\frac{(a1+a2)\left(-c1\, e^{a1\, x+\frac{y}{a1}}+c2\, e^{a2\, x+\frac{y}{a2}}\right)}{(a1-a2)\left(1+c1\, c2\, e^{(a1+a2)\, x+\left(\frac{1}{a1}+\frac{1}{a2}\right) y}\right)}$$

Out[136]=

$$-\frac{i\, a\left((c-i\, d)\, e^{(a-i\, b)\, x+\frac{y}{a-i\, b}}-(c+i\, d)\, e^{(a+i\, b)\, x+\frac{y}{a+i\, b}}\right)}{b\left(1+(c-i\, d)\,(c+i\, d)\, e^{2\, a\, x+\left(\frac{1}{a-i\, b}+\frac{1}{a+i\, b}\right) y}\right)}$$

Out[137]=

$$-\frac{2\, a\, e^{a\, x+\frac{a\, y}{a^2+b^2}}\left(d\, \mathrm{Cos}\left[b\, x-\frac{b\, y}{a^2+b^2}\right]+c\, \mathrm{Sin}\left[b\, x-\frac{b\, y}{a^2+b^2}\right]\right)}{b\left(1+c^2\, e^{2\, a\, x+\frac{2\, a\, y}{a^2+b^2}}+d^2\, e^{2\, a\, x+\frac{2\, a\, y}{a^2+b^2}}\right)}$$

Out[138]=

0

Это можно переобозначить так:

In[139]:=

```
V[x_, y_, a_, b_, C1_, C2_] = (a Sin[b (x - y/(a^2+b^2)) + C1])/(b Cosh[a (x + y/(a^2+b^2)) + C2]);
v = %
Simplify[D[v, x, y] - v/(1 + v^2) (2 D[v, x] × D[v, y] + 1 - v^2)]

VV[x_, y_, a_, b_, C1_, C2_] = a/(b Cosh[a (x + y/(a^2+b^2)) + C2]);
```

Out[140]=

$$\frac{a\, \mathrm{Sech}\left[C2+a\left(x+\frac{y}{a^2+b^2}\right)\right]\, \mathrm{Sin}\left[C1+b\left(x-\frac{y}{a^2+b^2}\right)\right]}{b}$$

Out[141]=

0

In[143]:=

```
Manipulate[
 Plot[{4 ArcTan[VV[x, y, a, b, 0, 0]], 4 ArcTan[V[x, y, a, b, 0, 0]]}, {x, -50, 50},
  PlotRange → {-1.5, 1.5},
  PlotPoints → 100,
  PlotStyle → {{Dashing[{0.01}], Red}, {Blue}},
  AxesLabel → {"x", "u"},
  ImageSize → 500,
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"}],
 {{a, 0.5}, -1, 1, Appearance → "Labeled"},
 {{b, 1.5}, -5, 5, Appearance → "Labeled"},
 {{y, 0}, -70, 70}
]
```

Out[143]=



In[144]:=

```
Plot3D[4 ArcTan[V[x, y, 0.5, 1.5, 0, 0]], {x, -20, 20}, {y, -20, 20},
 PlotRange → {-2, 2},
 PlotPoints → 100,
 Mesh → False,
 ColorFunction → "Aquamarine",
 AxesLabel → {"x", "y", "u"},
 ImageSize → 500,
 BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"}]
```

Out[144]=



В частности, если $a^2 + b^2 = 1$, то в лабораторных координатах получается стоячий бризер:

In[145]:=

```
V[x, y, 3/5, 4/5, 0, 0]
```

Out[145]=

$$\frac{3}{4}\,\text{Sech}\left[\frac{3\,(x+y)}{5}\right]\,\text{Sin}\left[\frac{4\,(x-y)}{5}\right]$$

In[146]:=

```
Plot3D[4 ArcTan[V[x, y, 3/5, 4/5, 0, 0]], {x, -20, 20}, {y, -20, 20},
 PlotRange → {-5, 5},
 PlotPoints → 50,
 Mesh → False,
 ColorFunction → "Aquamarine",
 AxesLabel → {"x", "y", "u"},
 ImageSize → 500,
 BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"}]
```

Out[146]=